

# Built-in Excel Functions II

► More examples of Excel Built-in functions:

- Mathematical & Trigonometric Functions

Most of these are standard functions you also find on a calculator:

**=SIN(x), =COS(x), =TAN(x),** (x is an angle in radians)

**=ACOS(x),** inverse cosine function

**=EXP(x),** exponential function

**=LN(x),** logarithm of x on base e=2,7182...

**=LOG(x),** logarithm of x on base 10

**=LOG(x,y),** logarithm of x on base y

**=COSH(x),** hyperbolic cosine function

**=ABS(x),** absolute value function

**=FACT(x),** factorial function

- Statistical Functions

These are functions you will encounter when you study probability and statistics. Examples are:

**=AVERAGE(A1:B7)** returns the arithmetic mean of A1:B7

**=MAX(A1:B7)** returns the largest number in A1:B7

**=COUNT(A1:B7)** returns the number of cells in range A1:B7 that contain numbers

**=SLOPE(A1:A7,B1:B7)**

**=INTERCEPT(A1:A7,B1:B7)**

return the slope (b) and the intercept (a) of a line  $y = a + b x$ , where A1:A7 contain values of y and B1:B7 contain the corresponding values of x.

**=VAR(A1:A7)** returns the variance of the set of numbers in cells A1:A7

- Text Functions

These are functions which manipulate text strings. Examples are:

**=EXACT(text1,text2)** , returns “TRUE” if text1=text2 and “FALSE” if text1≠text2 (case sensitive)

**=UPPER(text)**,

**=LOWER(text)**,

convert all characters of text to upper (lower) case


**=PROPER(text)**,

returns text in the “proper” style, that is the initial letter of every word in upper case and every other letter in lowercase

When you write text as argument of a function you always need to use quotation marks!

<b>=PROPER("to be or not be, thus is the question")</b>					
PROPER(text)					

To Be Or Not Be, Thus Is The Question					



- Information Functions

These are functions which return informations about the cell data or format. Examples are:

**=TYPE(A1)** returns a number which stands for the data type contained in cell A1: **1** = number, **2** = text, **4** = logical value, **16** = error value, **64** = array

**=ERROR.TYPE(A1)**

if cell A1 contains an error message, this function returns a number that stands for the type of error as follows: **1**= #NULL!,  
**2**= #DIV/0! , **3**= #VALUE!, **4**= #REF!,  
**5**= #NAME?,**6**= #NUM!,**7**= #N/A

**=ISODD(A1)**

returns “true” if A1 is an odd number and “false” otherwise

- There are various ways to make errors when using functions.

Excel will give the following **error messages**:

**#DIV/0!** ≡ division by zero

**#NAME?** ≡ a formula contains an undefined variable or function name, or the function syntax is not valid

**#N/A** ≡ value is not available, that is when formulae refer to cells which do not contain the appropriate data

**#NUM!** ≡ numerical overflow, e.g. SQRT(A1) for A1 is -5

**#VALUE!** ≡ invalid argument type, e.g. SQRT(A1) for A1 containing text

**#REF!** ≡ invalid cell reference

**circular error** ≡ a formula contains a reference to its own location

- Logical Functions

Last week we already saw some examples with the IF function.

Today we will see further examples of nested IF functions.

IF functions can also be used in conjunction with other logical functions, such as AND and OR.

IF structures are so important that many other Excel functions contain implicitly IF functions. Example:

**=SUMIF**(range,condition,sum\_range)

range       ≡ The range to be evaluated.


condition   ≡ A criterium which selects out certain values.

sum\_range   ≡ The range which will actually be summed.

It is optional. When omitted it corresponds to range.

**Examples:** `=SUMIF(B1:B10,"<10")` sums up all values in the cells B1:B10 which are smaller than 10.

`=SUMIF(A1:A10,"YES",B1:B10)` sums up all the values in the cells B1:B10 when YES is written in the corresponding cell of the range A1:A10.




The screenshot shows an Excel spreadsheet with the following data:

	A	B	C	D	E	F
1		5				
2		557				
3	YES	30				
4	*	10.77		21		
5	19	9				
6		123				
7	No	7		41		
8		38				
9	YES	11				
10		12				

The formula bar shows the formula `=SUMIF(A1:A10,"YES",B1:B10)` and the result 41 is displayed in cell D7.


Also Formats can contain an implicit IF-statement. This is called **conditional Formatting**. It is for instance useful when some cells only contain data under certain circumstances.

You can use conditional formatting by selecting on the Home tab of the Excel WS.




You have lots of options!

Suppose you want to highlight in a different colour all cells that contain numerical values less than a given value



	A	B	C	D	E	F	G	H	I
1	10347,00								
2	12089,00								
3	40000,00								
4	43000,00								
5	45890,00								
6	23000,00								
7									





The **AND**-function can be used to produce more complex tests. It returns the logical value TRUE if **all** conditions in its argument are true and the value FALSE otherwise.

Syntax: =AND(condition1,condition2,condition3,...)

**Examples:** =AND(A1>5,A2>5) returns TRUE if both A1 and A2 are greater than 5

The example from our previous lecture (see page 16) can be also written as:

=IF(AND(A1>-5, A1<=5) ,1 , 0 )

$$f(A1) = \begin{cases} 0 & \text{for } A1 \leq -5 \\ 1 & \text{for } -5 < A1 \leq 5 \\ 0 & \text{for } A1 > 5 \end{cases}$$

The **OR**-function can also be used to produce more complex tests. It returns the logical value TRUE if **at least one** condition in its argument is true.

Syntax: =OR(condition1,condition2,condition3,...)

**Examples:** =OR(A1>5, A2>5, B1>5, D11>5 )

Returns TRUE if any of the values in A1, A2,B1,D11 is greater than 5 and otherwise FALSE.

=IF(OR(A1<=-5, A1>5) ,0 , 1 )

Produces the same function  $f(A1)$  as the example in the previous page!

$$f(A1) = \begin{cases} 0 & \text{for } A1 \leq -5 \\ 1 & \text{for } -5 < A1 \leq 5 \\ 0 & \text{for } A1 > 5 \end{cases}$$

The **NOT**-function reverses the values of its logical argument, that is TRUE is changed into FALSE and vice versa.

Syntax: =NOT(condition)

**Examples:** If we look at all the ways in which we have been able to write the function f(A1) so far, we can always find alternative ways, by employing the NOT function:

=IF(A1>-5, IF(A1<=5,1,0) ,0)

=IF(A1>-5, IF(**NOT**(A1<=5),0,1) ,0)

=IF(**AND**(A1>-5,A1<=5),1,0)

=IF(**NOT**(**AND**(A1>-5,A1<=5)),0,1)

=IF(**OR**(A1<=-5,A1>5),0,1)

=IF(**NOT**(**OR**(A1<=-5,A1>5)),1,0)

All these functions do exactly the same, by using **different combinations** of logical functions!!

From the examples we can see that by combining logical functions we can produce **very complicated functions** . Let us see an example from a past exam paper (2007):

a) Write down the function that is produced by the following combination of Excel built-in functions:

=IF(x>=-3, IF(OR(x<=1,x>=3),SQRT((x-1)\*(x^2-9)), "complex"), "complex").

b) Compute the values of the function for x=-2.3 and x=67.

### How do we proceed?

we analyze each part of the function (each argument)

 we produce some sort of diagram of the function that shows what its value is for a given value of x

=IF( $x \geq -3$ , IF(OR( $x \leq 1, x \geq 3$ ), SQRT( $(x-1)(x^2-9)$ ), "complex"), "complex").


Every IF function has three arguments:

Condition:  $x \geq -3$

Value if condition is TRUE:

IF(OR( $x \leq 1, x \geq 3$ ), SQRT( $(x-1)(x^2-9)$ ), "complex")

Value if condition is FALSE: "complex"



The correct answer to the question would be:

$$\text{☞ } f(x) = \begin{cases} \text{complex,} & \text{for } x < -3 \\ \sqrt{(x-1)(x^2-9)}, & \text{for } -3 \leq x \leq 1 \\ \text{complex,} & \text{for } 1 < x < 3 \\ \sqrt{(x-1)(x^2-9)}, & \text{for } x \geq 3 \end{cases}$$

$$f(-2.3) = 3.4990... \quad f(67) = 543.7647...$$